```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the first Weyl algebra $A = A_1(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d,t], polynom=[t]):
```

the left A-module M finitely presented by the matrix defined by

```
> R := evalm([[t^2,t],[t*d+2,d]]);
```

$$R := \begin{bmatrix} t^2 & t \\ t\,d+2 & d \end{bmatrix}$$

and the left A-module M' finitely presented by the matrix defined by

```
> Rp := evalm([[t^2],[t*d+2]]);
```

$$Rp := \begin{bmatrix} t^2 \\ t\,d+2 \end{bmatrix}$$

Let us also consider the left A-homomorphism χ from M' to M induced by the matrix defined by

```
> W := evalm([[1,0]]);
```

$$\mathrm{W} := \begin{bmatrix} 1 & 0 \end{bmatrix}$$

We can check that χ is a left A-isomorphism:

```
> TestIso(Rp,R,W,A);
```

$$\mathit{true}$$

Since M is isomorphic to M', the direct sum of $A^{1\times 4}$ and N_1 and the direct sum of $A^{1\times 3}$ and N_2, where N_1 and N_2 are finitely presented by the formal adjoint of R and R', respectively, are isomorphic. The corresponding left A-isomorphism can be computed by the command *AuslanderEquivalence*:

```
> unprotect(O);
> O := AuslanderEquivalence(Rp,R,W,A,opt):
```

We obtain that the left A-module L finitely presented by the matrix defined by

```
> P := O[1];
```

$$P := \begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & t^2 & 1-t\,d \\ 0 & 0 & 0 & 0 & 0 & t & -d \end{bmatrix}$$

is isomorphic to the direct sum of $A^{1\times 4}$ and N_1, and the left A-module L' finitely presented by the matrix defined by

```
> Pp := O[2];
```

$$Pp := \begin{bmatrix} 0 & 0 & 0 & t^2 & 1-t\,d & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

is isomorphic to the direct sum of $A^{1\times 3}$ and N_2. Moreover, the matrix defined by

```
>  X := O[3];
```

$$X := \begin{bmatrix} 0 & -1 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 & 1 & 0 & 0 \\ 1 & -t^2 & -1+t\,d & 0 & 0 & 1 & 0 \\ 0 & -t & d & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & t^2 & 1-t\,d & -1 & 0 \\ 0 & 0 & 0 & t\,d+2 & -d^2 & 0 & -d \\ 0 & 0 & 0 & t^2 & 1-t\,d & 0 & -t \end{bmatrix}$$

induces a left A-homomorphism f from L to L' which is a left A-isomorphism:

```
>  TestIso(P,Pp,X,A);
```

$$true$$

The inverse f^{-1} of f is induced by the matrix defined by:

```
>  Xp := O[5];
```

$$Xp := \begin{bmatrix} -t^2 & -1+t\,d & 1 & 0 & 1 & 0 & 0 \\ -t\,d-2 & d^2 & 0 & d & 0 & 1 & 0 \\ -t^2 & -1+t\,d & 0 & t & 0 & 0 & 1 \\ -t\,d-1 & d^2 & 0 & d & 0 & 1 & 0 \\ -t^2 & t\,d & 0 & t & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & -1 & t^2 & 1-t\,d \\ 0 & 0 & 0 & 0 & 0 & t & -d \end{bmatrix}$$

We can check again that the above left A-homomorphism is an isomorphism:

```
>  TestIso(Pp,P,Xp,A);
```

$$true$$

If P = (0 P2), where P2 is defined by

```
>  P2 := submatrix(P,1..3,2..7);
```

$$P2 := \begin{bmatrix} 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & t^2 & 1-t\,d \\ 0 & 0 & 0 & 0 & t & -d \end{bmatrix}$$

and P' = (0, P2'), where P2' is defined by

```
>  P2p := submatrix(Pp,1..3,2..7);
```

$$P2p := \begin{bmatrix} 0 & 0 & t^2 & 1-t\,d & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

L2 is the left A-module finitely presented by P2 and L2' is the left A-module finitely presented by P2', then L2 is isomorphic to L2'. Let us compute the corresponding left A-isomorphism f2.

```
> Y := Cancellation(P2p,X,A);
```

$$Y := \begin{bmatrix} 0 & -1 & 0 & 1 & 0 & 0 \\ -1 & -1+t\,d & 1-t^2 & 0 & 1 & 0 \\ 0 & d & -t & 0 & 0 & 1 \\ 0 & 0 & t^2 & 1-t\,d & -1 & 0 \\ 0 & 0 & t\,d+2 & -d^2 & 0 & -d \\ 0 & 0 & t^2 & 1-t\,d & 0 & -t \end{bmatrix}$$

We obtain that the left A-homomorphism f2 from L2 to L2' induced by Y is a left A-isomorphism.

```
> TestIso(P2,P2p,Y,A);
```

$$true$$

If P2 = (0 P3) and P2' = (0 P3'), where

```
> P3 := submatrix(P2,1..3,2..6);
```

$$P3 := \begin{bmatrix} 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & t^2 & 1-t\,d \\ 0 & 0 & 0 & t & -d \end{bmatrix}$$

```
> P3p := submatrix(P2p,1..3,2..6);
```

$$P3p := \begin{bmatrix} 0 & t^2 & 1-t\,d & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

L3 is the left A-module finitely presented by P3 and L3' is the left A-module finitely presented by P3', then L3 is isomorphic to L3'. Let us compute the corresponding left A-isomorphism f3.

```
> Z := Cancellation(P3p,Y,A);
```

$$Z := \begin{bmatrix} 1 & 1-t^2 & -2+t\,d & 1 & 0 \\ 0 & -t & d & 0 & 1 \\ 0 & t^2 & 1-t\,d & -1 & 0 \\ 0 & t\,d+2 & -d^2 & 0 & -d \\ 0 & t^2 & 1-t\,d & 0 & -t \end{bmatrix}$$

We obtain that the left A-homomorphism f3 from L3 to L3' induced by Z is an isomorphism. Let us check that f3 is a left A-isomorphism.

```
> TestIso(P3,P3p,Z,A);
```

$$true$$

Finally, if P3 = (0 P4), where

```
> P4 := submatrix(P3p,1..3,2..5);
```

$$P4 := \begin{bmatrix} t^2 & 1-t\,d & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

P3' = (0 P4'), where

```
> P4p := submatrix(P3,1..3,2..5);
```

$$P4p := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & t^2 & 1 - t\,d \\ 0 & 0 & t & -d \end{bmatrix}$$

L4 is the left A-module finitely presented by P4, and L4' is the left A-module finitely presented by P4', then we cannot deduce a left A-isomorphism f4 from L4 to L4' from f3 since $rank_A(\text{L4'}) = 1$:

```
>  U := Cancellation(P4p,Z,A);
```

```
Error, (in Stafford/ReductionOfUnimodularElement) expecting that the
rank of the left module presented by the first matrix is at least 2.
```